Laborator 6: Puncte de echilibru. Stabilitate

In cadrul acestui laborator sunt prezentate instructiunile necesare studiului calitativ al solutiilor in jurul punctelor de echilibru in cazul ecuatiilor scalare autonome si a sistemelor planare de ecuatii autonome.

Ecuatii scalare autonome

Ecuatiile scalare autonome sunt de forma:

$x' = f(x)$

Se considera ecuatia autonoma

$x' = x\left(1 - x^2\right)$

```
> with(DEtools): with(plots):
```

construim functia din membrul drept:

```
> f:=x->x*(1-x^2);
```

$$f := x \to x\left(1 - x^2\right)$$

```
> ec:=diff(x(t),t)=f(x(t));
```

$$ec := \frac{d}{dt}x(t) = x(t)\left(1 - x(t)^2\right)$$

Punctele de echilibru sunt solutiile reale ale ecuatiei

$f(x) = 0$

```
> pct_ech:=solve(f(x)=0,x);
```

$$pct\_ech := 0, 1, -1$$

Pentru studiul stabilitatii sunt doua metode, fie aplicam Teorema stabilitatii in prima aproximatie, fie prin metoda grafica (analiza portretului fazic).

1. Aplicarea Teoremei stabilitatii in prima aproximatie - se evalueaza f ' in punctele de echilibru

```
> pct_ech[1];
```

$$0$$

```
> D(f)(pct_ech[1]);
```

$$1$$

Observam ca $f'(0) = 1 > 0$, deci punctul de echilibru 0 este **instabil**.

In mod analog procedam si in cazul celorlalte puncte de echilibru:

```
> D(f)(pct_ech[2]);
```

$$-2$$

```
> D(f)(pct_ech[3]);
```

$$-2$$

Observam ca $f'(1) = f'(-1) = 2 < 0$, deci punctele de echilibru 1 si -1 sunt **local asimptotic stabile**

2. Metoda grafica (analiza portretului fazic)

In cazul ecuatiilor scalare autonome MAPLE nu are o comanda de generare a portretului fazic, stabilitatea solutiilor de echilibru se face prin analizarea campului de directii si reprezentarea grafica a solutiilor reprezentative utilizand comanda **DEplot**. In cazul ecuatiei noastre solutiile reprezentative sunt solutiile ce satisfac conditii initiale in 0 mai mici decat -1, solutii ce satisfac conditii initiale in 0 cuprinse intre -1 si 0, solutii ce satisfac conditii initiale in 0 cuprinse intre 0 si 1, solutii ce satisfac conditii initiale in 0 mai mari decat 1:

```
> DEplot(ec,x(t), t=-2..2,
[[x(0)=-3],[x(0)=-2],[x(0)=-1],[x(0)=-1/2],[x(0)=-1/3],[x(0)=0
],[x(0)=1/3],[x(0)=1/2],[x(0)=1],[x(0)=2],[x(0)=3]]);
```



Sisteme liniare

Se considera sistemul liniar de doua ecuatii diferentiale:

$x' = x + y$

$y' = x - y$

Sistemul dat se introduce definind cele doua ecuatii prin intermediul procedurii **diff**

```
> restart;
> with(DEtools): with(plots):with(linalg):
> ec1:=diff(x(t),t)=x(t)+y(t);
```

$$ec1 := \frac{d}{dt}x(t) = x(t) + y(t)$$

```
> ec2:=diff(y(t),t)=x(t)-y(t);
```

$$ec2 := \frac{d}{dt}y(t) = x(t) - y(t)$$

```
> sist:=ec1,ec2;
```

$$sist := \frac{d}{dt}x(t) = x(t) + y(t), \frac{d}{dt}y(t) = x(t) - y(t)$$

Construim matricea sistemului dupa care calculam valorile proprii corespunzatoare:

```
> A:=matrix([[1,1],[1,-1]]);
```

$$A := \begin{bmatrix} 1 & 1 \\ 1 & -1 \end{bmatrix}$$

```
> eigenvals(A);
```

$$\sqrt{2}, -\sqrt{2}$$

Cum prima valoare proprie este strict pozitiva iar a doua este strict negativa atunci punctul de echilibru (0,0) este punct de echilibru **instabil de tip sa**.
Pentru vizualizarea dinamicii generate de acest sistem se reprezinta protretul fazic impreuna cu cateva orbite utilizand comanda `DEplot`

```
>
cond_in2:=[x(0)=0,y(0)=i]$i=1..5,[x(0)=-i,y(0)=0]$i=1..5,[x(0)
=0,y(0)=-i]$i=1..5,[x(0)=i,y(0)=0]$i=1..5;
```

$cond\_in2 := [x(0) = 0, y(0) = 1], [x(0) = 0, y(0) = 2], [x(0) = 0, y(0) = 3], [x(0) = 0, y(0) = 4], [x(0) = 0, y(0) = 5], [x(0) = -1, y(0) = 0], [x(0) = -2, y(0) = 0], [x(0) = -3, y(0) = 0], [x(0) = -4, y(0) = 0], [x(0) = -5, y(0) = 0], [x(0) = 0, y(0) = -1], [x(0) = 0, y(0) = -2], [x(0) = 0, y(0) = -3], [x(0) = 0, y(0) = -4], [x(0) = 0, y(0) = -5], [x(0) = 1, y(0) = 0], [x(0) = 2, y(0) = 0], [x(0) = 3, y(0) = 0], [x(0) = 4, y(0) = 0], [x(0) = 5, y(0) = 0]$

```
> DEplot([sist],[x(t),y(t)],t=-5..5,x=-6..6,y=-6..6,[cond_in2],
arrows=medium, linecolor=blue,stepsize=0.1);
```



Sisteme neliniare. Stabilitatea puntelor de echilibru

Sa consideram urmatorul sistem neliniar:

$$x' = x\left(1 - \frac{x}{2} - y\right)$$
$$y' = y\left(x - 1 - \frac{y}{2}\right)$$

Pentru determinarea punctelor de echilibru definim functiile din membrul drept al fiecarei ecuatii

```
> f1:=(x,y)->x*(1-x/2-y);
```

$$f1 := (x, y) \rightarrow x\left(1 - \frac{1}{2}x - y\right)$$

```
> f2:=(x,y)->y*(x-1-y/2);
```

$$f2 := (x, y) \rightarrow y\left(x - 1 - \frac{1}{2}y\right)$$

```
> ec1:=diff(x(t),t)=f1(x(t),y(t));
```

$$ec1 := \frac{d}{dt}x(t) = x(t)\left(1 - \frac{1}{2}x(t) - y(t)\right)$$

```
> ec2:=diff(y(t),t)=f2(x(t),y(t));
```

$$ec2 := \frac{d}{dt}y(t) = y(t)\left(x(t) - 1 - \frac{1}{2}y(t)\right)$$

```
> sist2:=ec1,ec2;
```

$$sist2 := \frac{d}{dt}x(t) = x(t)\left(1 - \frac{1}{2}x(t) - y(t)\right), \frac{d}{dt}y(t) = y(t)\left(x(t) - 1 - \frac{1}{2}y(t)\right)$$

Punctele de echilibru sunt solutiile reale ale sistemului

$$f1(x, y) = 0$$
$$f2(x, y) = 0$$

```
> PctEch:=solve({f1(x,y)=0,f2(x,y)=0},{x,y});
```

$$PctEch := \{x = 0, y = 0\}, \{x = 0, y = -2\}, \{x = 2, y = 0\}, \left\{x = \frac{6}{5}, y = \frac{2}{5}\right\}$$

Se observa ca acest sistem are patru puncte de echilibru. Pentru studiul stabilitatii acestora trebuiesc determinate valorile proprii ale jacobianului calculat in aceste puncte de echilibru.

```
> PctEch[1,1];PctEch[1,2];
```

$$x = 0$$

$$y = 0$$

```
> J:=jacobian([f1(x,y),f2(x,y)],[x,y]);
```

$$J := \begin{bmatrix} 1 - x - y & -x \\ y & x - 1 - y \end{bmatrix}$$

Matricea sistemului liniarizat pentru punctul de echilibru (0,0) este::

```
> A1:=subs(PctEch[1,1],PctEch[1,2],eval(J));
```

$$A1 := \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$$

```
> eigenvals(A1);
```

$$1, -1$$

Cum prima valoare proprie este 1 iar a doua este -1 atunci punctul de echilibru (0,0) este punct de echilibru **instabil de tip sa**.

Procedam similar pentru celelalte puncte de echilibru.

```
> PctEch[2,1];PctEch[2,2];
```

$$x = 0$$

$$y = -2$$

```
> A2:=subs(PctEch[2,1],PctEch[2,2],eval(J));
```

$$A2 := \begin{bmatrix} 3 & 0 \\ -2 & 1 \end{bmatrix}$$

```
> eigenvals(A2);
```

$$3, 1$$

In acest caz punctul de echilibru (0,-2) este un punct de echilbru **instabil de tip nod**.

```
> PctEch[3,1];PctEch[3,2];
```

$$x = 2$$

$$y = 0$$

```
> A3:=subs(PctEch[3,1],PctEch[3,2],eval(J));
```

$$A3 := \begin{bmatrix} -1 & -2 \\ 0 & 1 \end{bmatrix}$$

```
> eigenvals(A3);
```

$$-1, 1$$

Si in acest caz punctul de echilibru (0,-2) este un punct de echilbru **instabil de tip sa**.

```
> PctEch[4,1];PctEch[4,2];
```

$$x = \frac{6}{5}$$

$$y = \frac{2}{5}$$

```
> A4:=subs(PctEch[4,1],PctEch[4,2],eval(J));
```

$$A4 := \begin{bmatrix} -\frac{3}{5} & -\frac{6}{5} \\ \frac{2}{5} & -\frac{1}{5} \end{bmatrix}$$

```
> eigenvals(A4);
```

$$-\frac{2}{5} + \frac{1}{5} I\sqrt{11}, -\frac{2}{5} - \frac{1}{5} I\sqrt{11}$$

In acest caz avem valori proprii complexe conjugate a caror parte reala este -2/5, deci punctul de echilibru (6/5,2/5) este un puct de echilibru **asimptotic stabil de tip focus**.

Pentru reprezentarea portretului fazic trebuie aleasa o fereastra ce contine toate punctele de echilibru, in cazul acestui exemplu putem alege `x=-3..3, y=-3..3`, si apoi fixata o lista de conditii pentru a reprezenta cateva orbite.

```
>
condin:=[x(0)=-1,y(0)=1],[x(0)=-0.5,y(0)=1],[x(0)=1,y(0)=1],[x(0)=1,y(0)=3],[x(0)=2,y(0)=0.5],[x(0)=-1,y(0)=-1],[x(0)=-0.5,y(0)=-1],[x(0)=-1,y(0)=-2.5],[x(0)=1,y(0)=-1],[x(0)=1.5,y(0)=-1],[x(0)=1,y(0)=-2.5];
```

$$condin := [x(0) = -1, y(0) = 1], [x(0) = -0.5, y(0) = 1], [x(0) = 1, y(0) = 1], [x(0) = 1, y(0) = 3], [x(0) = 2, y(0) = 0.5], [x(0) = -1, y(0) = -1], [x(0) = -0.5, y(0) = -1], [x(0) = -1, y(0) = -2.5], [x(0) = 1, y(0) = -1], [x(0) = 1.5, y(0) = -1], [x(0) = 1, y(0) = -2.5]$$

```
> DEplot([sist2],[x(t),y(t)],t=-10..10,x=-3..3,y=-3..3,
[condin],linecolor=blue,stepsize=0.1);
```